Computers & Geosciences Vol. 17, No. 5, pp. 669–678, 1991
Printed in Great Britain

0098-3004/91 $3.00 + 0.00
Pergamon Press plc

GEOMAGNETIC, GEOCENTRIC, AND GEODETIC COORDINATE TRANSFORMATIONS

CHARLES E. BARTON and CHRIS Z. TARLOWSKI
Bureau of Mineral Resources, P.O. Box 378, Canberra, ACT 2601, Australia

(Received 13 September 1990; accepted 28 December 1990)

Abstract—A requirement in geomagnetism is to make transformations between geocentric, geodetic (spheroidal), and geomagnetic coordinate systems. A few of these transformations may be included in textbooks, but we have never noted them all presented. This paper provides derivations and subroutines for conversion of spatial coordinates and Cartesian magnetic field components between any combinations of these three coordinate systems.

Key Words: Coordinate transformations, Geomagnetism, Geocentric, Geodetic, Geodesy.

INTRODUCTION

Satellite positions and observations usually are expressed in geocentric coordinates. Most geophysical observations made at, or near, the surface of the Earth are expressed in geodetic coordinates references to an ellipsoidal (spheroidal) Earth. Many geomagnetic phenomena are controlled by the Earth's geomagnetic dipole and are best treated in geomagnetic coordinates. If, like us, you have needed to transform spatial coordinates (and magnetic field components) between these three coordinate systems, then you probably share our frustration in never being able to locate the required equations in the literature. This paper provides equations and FORTRAN-77 subroutines for performing any of the six tranformations between these coordinate systems, and includes appropriate transformations for Cartesian components of the magnetic field.

Equations for converting from geocentric to geomagnetic coordinates are presented in many basic textbooks on geomagnetism (e.g. Chapman and Bartels, 1940; Matsushita and Campbell, 1967). Likewise, equations for converting from geodetic to geocentric coordinates are given (e.g. Zmuda, 1971—as used in most computer programs for evaluating spherical harmonic models of the geomagnetic field, such as IGRF). However, the reverse transformations are seldom documented, particularly the conversion from geocentric to geodetic coordinates—perhaps because the equations are inelegant and require the solution of a quartic equation. Some of the widely used subroutines place restrictions on which quadrant(s) can be used and on the sign of the geodetic height (altitude), whereas those presented here apply in all situations provided sensible values of the parameters are entered. Angles are given in degrees unless stated otherwise.

GEOCENTRIC COORDINATES

Geocentric coordinates are spherical polar coordinates with origin at the center of mass of the Earth and polar axis along the northwards rotation axis of the Earth (ON), as illustrated in Figure 1. Longitude is referred to the Greenwich meridian (NGS). A point in space is designated by colatitude (θ), longitude (ϕ) measured eastwards from the Greenwich meridian, and radial distance (r) from the origin. The geocentric latitude of P is angle λ, equal to $90 - \theta$. Unlike the geodetic and geomagnetic coordinate sytems, the geocentric system does not require specification of the particular earth model that is used for reference.

GEODETIC COORDINATES

An oblate ellipsoid approximating to the surface of the Earth is used as a reference surface (Fig. 1). The axis of the ellipsoid is the rotation axis of the Earth (ON), and the ellipsoid is defined by its major semiaxis, a, and flattening, f. The minor semiaxis, b, is derived from

$$f = (a - b)/a$$

and ellipticity, e, is given by

$$e^2 = (a^2 - b^2)/a^2. \tag{1}$$

A point P is designated by its geodetic colatitute, ψ, longitude, ϕ, and perpendicular (vertical) height, h, above the ellipsoid. (In practice the local vertical at any point, as measured by a plumb-line, generally will not be exactly parallel to the normal to the ellipsoid.) The geodetic latitude of P is the angle β, equal to $90 - \psi$. Examples of some commonly used reference ellipsoids are given in Table 1.



Figure 1. Geocentric and geodetic coordinates illustrated on reference ellipsoid with major semiaxis, a, and minor semiaxis, b. NS is rotation axis of Earth. PQ is normal from point P to surface of ellipsoid. NQM is meridian plane containing P. Geocentric coordinates of P are colatitude, θ (or latitude, λ), longitude, ϕ, and radial distance, r, from center of Earth (O). Geodetic coordinates of P are colatitude, ψ (or latitude, β), longitude, ϕ, and vertical height, h, above ellipsoid ($=PQ$). Geocentric and geodetic longitudes are same, measured eastwards from Greenwich meridian (NGS).

Table 1. Reference ellipsoids

Name	a (km)	$1/f$	b (km)
Clarke ellipsoid (1866)	6378.206	294.98	6356.58350
International ellipsoid (IAG, Madrid, 1924)	6378.388	297.00	6356.91195
Krassovsky ellipsoid (1942)	6378.245	298.30	6356.86302
IAU ellipsoid (Hamburg, 1964)	6378.160	298.25	6356.77472
Australian national ellipsoid (1966)	6378.160	298.25	6356.77472
South American ellipsoid (1969)	6378.160	298.25	6356.77472
World Geodetic System ellipsoid (1972)	6378.135	298.26	6356.75052

a—Major semiaxis; f—flattening; b—minor semiaxis.

GEOMAGNETIC COORDINATES

The geomagnetic coordinate system (Schmidt, 1918) is a spherical polar coordinate frame with polar axis coincident with the north-pointing, centered-dipole (geomagnetic) axis of the Earth. The zero-longitude meridian is defined as the geomagnetic meridian containing the south geographic pole. Geomagnetic coordinates are distinct from magnetic coordinates (see Matsushitsa and Campbell, 1967, Appendix 1).

The position of the geomagnetic axis is subject to secular variation, hence when specifying geomagnetic coordinates it is necessary to state the coordinates of the north (or south) geomagnetic pole—usually in geocentric coordinates, which are derived readily from the spherical harmonic Gauss coefficients. Compilations of geomagnetic dipole coordinates have been published by many authors (e.g. Voppel, 1985; Barton, 1989). A widely used standard for the geomagnetic axis is with the north geomagnetic pole at colatitute 11.5°, longitude 291.0°E (e.g. Chapman and Bartels, 1940, table A; Voppel, 1985). Some other values are listed in Table 2.

Table 2. Geocentric coordinates of north geomagnetic pole

Source	Colatitude (°)	Longitude (°E)
Matsushita and Campbell (1967)	11.7	291.0
MGST481-1 @ 1980.0	11.198	289.245
DGRF 1980 @ 1980.0	11.194	289.095
IGRF 1985 @ 1985.0	11.018	289.095

THE SUBROUTINES

The following FORTRAN subroutines for performing the transformations are listed in Appendix 2:

DTOCB—geodetic to geocentric

CTODB—geocentric to geodetic

CTOMB—geocentric to geomagnetic

MTOCB—geomagnetic to geocentric.

Letters "C", "D", and "M" denote geocentric, geodetic, and geomagnetic respectively; "B" indicates that the routines include a conversion for the Cartesian magnetic field components—northwards (BN), eastwards (BE), and vertically downwards (BV). It is a simple matter to remove the code for the field components if coordinate transformations alone are required.

Derivations of the equations used are given in Appendix 1. Conversion between geomagnetic and geodetic frames is accomplished by converting to and from geocentric coordinates as an intermediate step.

Double precision (real*8) has been used throughout. The subroutines for converting between geocentric and geodetic coordinates are coded to reduce rounding errors arising from taking powers of large numbers. For the spherical rotations (subroutines CTOMB and MTOCB) single precison would be sufficiently accurate for many purposes.

Spatial coordinates are expressed as colatitude, longitude (positive east of the reference meridian), and either radial distance in kilometers from the center of the Earth (geocentric and geomagnetic frames) or vertical height in kilometers above the reference ellipsoid (geodetic frame). Colatitude, or polar distance, is permitted values within the range 0–180°. Longitude can have any real values. In order to make the subroutine calls symmetrical we have selected to include the coordinates and components that are unchanged by the transformations. The "vertical" direction for any point lies along the normal to the reference surface passing through the point. In practice the local vertical, as measured by a plumb-line, generally will not be exactly parallel to this normal.

Acknowledgments—We thank Phil McFadden for his helpful comments. This paper is published with the permission of the Director, Bureau of Mineral Resources, Geology and Geophysics.

REFERENCES

Barton, C. E., 1989, Geomagnetic secular variation: direction and intensity, *in* James, D. E., ed., The encyclopedia of solid earth geophysics: Van Nostrand-Reinhold, New York, p. 560–577.

Chapman, S., and Bartels, J., 1940, Geomagnetism: Oxford Univ. Press, 2 volumes, 1049 p.

Matsushita, S., and Campbell W. H., 1967, Physics of geomagnetic phenomena: Academic Press, New York, 2 volumes, 1398 p.

Schmidt, A., 1918, Geomagnetische koordinaten: Archiv. des Erdmagnetismus, Potsdam, Heft 3, p. 14.

Voppel, D., 1985, Observation and description of the main geomagnetic field and its secular variation, *in* Fuchs, K., and Soffel, H., eds., Landolt-Börnstein, numerical data and functional relationships in science and technology, New Series, v. 2b: Springer-Verlag, Berlin, p. 126–180.

Zmuda, A. J., 1971, World Magnetic Survey 1957–1969: IAGA Bull. 28, IUGG Publ. Office, Paris, 206 p.

APPENDIX 1

Derivation of Equations

We make use of the equations for solving a spherical triangle with included angles A, B, C (in radians) facing sides a, b, c. For a sphere of unit radius the side lengths are equal to the respective angles in radians subtended at the center of the sphere.

Cosine rule:

$$\cos a = \cos b \cos c + \sin b \sin c \cos A \qquad \text{(A1)}$$

Sine rule:

$$\frac{\sin A}{\sin a} = \frac{\sin B}{\sin b} = \frac{\sin C}{\sin c}. \qquad \text{(A2)}$$

For a spherical triangle with a right-angle at A:

$$\cos a = \cos b \cos c. \qquad \text{(A3)}$$

Geodetic to geocentric conversion

The equations derived here are less complicated than those quoted by Zmuda (1971), which are used in the subroutines provided by the U.S. Geological Survey and the British Geological Survey for evaluating spherical harmonic models of the field such as IGRF. Furthermore, our equations give the correct result for points below the surface of the reference ellipsoid (negative heights).

Refer to Figure A1. Geocentric and geodetic longitudes are the same, so the problem is reduced to 2 dimensions. Given the geodetic coordinates (h, ψ) of point P, we wish to determine its geocentric coordinates (r, θ) with respect to the reference ellipsoid with major semiaxis, a, and minor semiaxis, b. We work in Cartesian coordinates, where the "y" axis is the polar axis, the "x" axis lies in the equatorial plane. $Q(X, Y)$ is the point on the reference ellipsoid vertically below $P(X_p, Y_p)$. The geodetic latitude, $\beta = 90 - \psi$.

Because Q lies on the ellipsoid,

$$X^2/a^2 + Y^2/b^2 = 1. \qquad \text{(A4)}$$

The normal to the ellipsoid through Q is

$$x\, a^2/X - y\, b^2/Y = a^2 - b^2$$

which has slope

$$\tan \beta = \mathrm{d}y/\mathrm{d}x = (a^2/b^2)(Y/X). \qquad \text{(A5)}$$

From Equations (5) and (6),

$$X = a^2/\sqrt{(a^2 + b^2 \tan \beta)}$$

and

$$Y = (b^2/a^2) \tan(\beta)\, X$$

hence

$$X_p = X + h \cos \beta; \quad Y_p = Y + h \sin \beta.$$

The geocentric radial distance, r, and geocentric colatitude are determined from

$$r^2 = X_p^2 + Y_p^2; \quad \cos \theta = Y_p/r.$$

To rotate the Cartesian components of the field, we determine the angle, α, between the geocentric and geodetic verticals

$$\alpha = \beta - \lambda = \beta - 90 + \theta. \qquad \text{(A6)}$$



Figure A1. Section of meridian plane containing point $P\ (X_p,\ Y_p)$, shown in Cartesian coordinates. y-Axis is rotation axis of Earth; x-axis lies in equatorial plane. PQ is normal to ellipsoid passing through P. Geocentric colatitude and radial distance of P are θ and r respectively. Geodetic coordinates of P are latitude, β, and vertical height, h. Colatitude of P is ψ (i.e. $90-\beta$).



Figure A2. Illustration of point P (colatitude θ, longitude ϕ) on surface of spherical Earth. N is north geographic pole; M is north geomagnetic pole (colatitude θ_p, longitude ϕ_p); NG is Greenwich meridian. Geomagnetic coordinates of P (i.e. with respect to polar axis OM) are colatitude θ_m and longitude ϕ_m. Angular distances are shown on figure.

Then for both northern and southern hemispheres

$$BN_c = BN_d \cos\alpha - BV_d \sin\alpha$$

$$BV_c = BV_d \cos\alpha + BN_d \sin\alpha$$

and

$$BE_d = BE_c$$

where c and d denote geocentric and geodetic respectively.

Geocentric to geodetic conversion

Refer again to Figure A1. Given the geocentric coordinates (r, θ) of point P, we wish to determine the geodetic coordinates (h, ψ). The Cartesian coordinates of P are

$$X_p = r\sin(\theta); \quad Y_p = r\cos(\theta).$$

Because Q lies on the ellipsoid,

$$X^2/a^2 + Y^2/b^2 = 1 \tag{A7}$$

and P lies on the normal at Q,

$$X_p a^2/X - Y_p b^2/Y = a^2 - b^2 \tag{A8}$$

it follows that

$$A_0 + A_1 X + A_2 X^2 + A_3 X^3 + A_4 X^4 = 0 \tag{A9}$$

where

$$A_0 = -a^2 X_p^2$$
$$A_1 = 2a^2 e^2 X_p$$
$$A_2 = X_p^2 + Y_p^2 b^2/a^2 - a^2 e^4$$
$$A_3 = -2e^2 X_p$$
$$A_4 = e^4.$$

Equation (A9) can be solved efficiently and safely by using Newton's method, taking as the initial approximation for the root:

$$X_i = 0.5\,(a + b)\sin\theta.$$

The coefficients of Equation (A9) are real hence the roots are either real or occur as pairs of conjugate complex numbers. In our situation one root is positive, one is negative, and the remaining two are complex. It is evident from Figure A1 that for any position of P this starting value must be sufficiently close to the required (positive) root that Newton's method will converge quickly.

The constant EPS in subroutine CTOGB determines how accurately the root of the quadratic is determined. (In the subroutine the value of EPS is determined by the precision of the computer being used.) The iteration to determine the root is stopped when successive approximations differ by less than EPS. Having determined the root, X, the value of Y is obtained from Equation (A7). The height above the ellipsoid is then

$$h = \sqrt{[(X - X_p)^2 + (Y - Y_p)^2]}$$

The negative square root is taken if $Y > Y_p$. The geodetic latitude, β, is determined from Equation (A5).

The equations hold for points in the northern hemisphere. Points in the southern hemisphere are treated in terms of their mirror images in the northern hemisphere.

To rotate the Cartesian components of the field, we determine the angle between the geocentric and geodetic verticals, α, from Equation (A6). Then for the northern hemisphere

$$\mathrm{BN_d} = \mathrm{BN_c}\cos\alpha + \mathrm{BV_c}\sin\alpha$$

$$\mathrm{BV_d} = \mathrm{BV_c}\cos\alpha - \mathrm{BN_c}\sin\alpha$$

and for the southern hemisphere

$$\mathrm{BN_d} = \mathrm{BN_c}\cos\alpha - \mathrm{BV_c}\sin\alpha$$

$$\mathrm{BV_d} = \mathrm{BV_c}\cos\alpha + \mathrm{BN_c}\sin\alpha.$$

in both situations

$$\mathrm{BE_d} = \mathrm{BE_c}.$$

Geocentric to geomagnetic conversion

Refer to Figure A2. Given the geocentric colatitude and longitude of the north geomagnetic pole (θ_p, ϕ_p) and the geocentric colatitude and longitude (θ, ϕ) of P, we need to determine the geomagnetic colatitude and longitude (θ_m, ϕ_m) of P. Applying the cosine rule (Eq. [(A1)] to spherical triangle NMP.

$$\cos\theta_m = \cos\theta_p\cos\theta + \sin\theta_p\sin\theta\cos(\phi - \phi_p).$$

Taking the inverse cosine gives θ_m in the correct quadrant.

Applying the sine rule [Eq. (A2)],

$$\sin\gamma = \sin\theta\sin(\phi - \phi_p)/\sin\theta_p$$

whence

$$\phi_m = 180 - \gamma = \sin^{-1}[\sin\theta\sin(\phi - \phi_p)/\sin\theta_p].$$

This returns ϕ_m in the first and fourth quadrants only. Equation (A3) is used to test for the obtuse angle situation of the spherical triangle:

if

$$\cos\theta > \cos\theta_p\cos\theta_m$$

then

$$\phi_m = 180 - \sin^{-1}[\sin\theta\sin(\phi - \phi_p)/\sin\theta_p].$$

To rotate the Cartesian components of the field, we determine the angle between the geocentric and geomagnetic "north" by applying the sine rule to triangle NMP,

$$\alpha = \sin^{-1}[\sin\theta_p\sin(\phi - \phi_p)/\sin\theta_m] \tag{A10}$$

and test to determine the obtuse angle situation:

if

$$\cos\theta_p < \cos\theta\cos\theta_m$$

then

$$\alpha = 180 - \sin^{-1}[\sin\theta_p\sin(\phi - \phi_p)/\sin\theta_m]. \tag{A11}$$

Rotating the field components we get:

$$\mathrm{BN_m} = \mathrm{BN_c}\cos\alpha - \mathrm{BE_c}\sin\alpha$$

$$\mathrm{BE_m} = \mathrm{BE_c}\cos\alpha + \mathrm{BN_c}\sin\alpha$$

and

$$\mathrm{BV_m} = \mathrm{BV_c}$$

where suffixes m and c denote geomagnetic and geocentric respectively.

Geomagnetic to geocentric conversion

Refer to Figure A2. Given the geomagnetic pole (θ_p, ϕ_p) and the geomagnetic coordinates (θ_m, ϕ_m), we now wish to determine the geocentric colatitude and longitude (θ, ϕ) of P. Proceeding in much the same way as for the reverse transformation, we get

$$\cos\theta = \cos\theta_p\cos\theta_m + \sin\theta_p\sin\theta_m\cos\gamma.$$

Taking the inverse cosine gives the geocentric colatitude in the correct quadrant.

By the sine rule,

$$\sin(\phi - \phi_p) = \sin\theta_m \sin\gamma/\sin\theta$$
$$= \sin\theta_m \sin\phi_m/\sin\theta$$

whence

$$\phi = \phi_p + \sin^{-1}[\sin\theta_m \sin\phi_m/\sin\theta].$$

This returns ϕ in the first and fourth quadrants only. Testing for the obtuse angle situation,

if

$$\cos\theta_m < \cos\theta_p \cos\theta$$

then

$$\phi = \phi_p + 180 - \sin^{-1}[\sin\theta_m \sin\phi_m/\sin\theta].$$

To rotate the Cartesian components of the field find angle α, from Equations (A10) or (A11), whence

$$BN_c = BN_m \cos\alpha + BE_m \sin\alpha$$
$$BE_c = BE_m \cos\alpha - BN_m \sin\alpha$$

and

$$BV_c = BV_m.$$

APPENDIX 2

Subroutine Listings

```
c-----------------------------------------------------------------------
      subroutine DTOCB(CLATD,ELOND,ALT,BXD,BYD,BZD,
     +                  CLATC,ELONC, R, BXC,BYC,BZC)
c
c Geodetic to geocentric conversion for coordinates and field
c I/O double precision (real*8)
c Input (geodetic)
c    CLATD= colatitude (deg), range 0-180
c    ELOND= longitude  (degE)
c    ALT  = height above spheroid (km), may be negative
c    BXD  = north component of field
c    BYD  = east  component of field
c    BZD  = vert  component of field
c Output (geocentric)
c    CLATC= colatitude (deg)
c    ELONC= longitude  (degE)
c    R    = radial distance from geocentre (km)
c    BXC  = north component of field
c    BYC  = east  component of field
c    BZC  = vert  component of field
c
      implicit real*8 (A-H,O-Z)
      rpd=datan(1.d0)/45.d0
      A = 6378.135d0       ! semi major axis, World spheroid, WGS 1972
      B = 6356.750d0       ! semi minor axis    "       "

      if(CLATD.lt.0.d0.or.CLATD.gt.180.d0) then       ! quit
        write(*,*)'Colatitude out of range in subroutine DTOCB'
        stop
      endif
c
c Convert to geocentric coordinates
c
      AA   = A*A
      BB   = B*B
      B2A2= (B/A)**2
      BETA= (90.d0-CLATD)*rpd             ! geodetic latitude (rad)
      TANB= dtan(BETA)
      X = A/dsqrt(1.d0+B2A2*TANB*TANB)
      Y = B2A2*TANB*X
      XP= X + ALT*dcos(BETA)
      YP= Y + ALT*dsin(BETA)
      R = dsqrt(XP*XP+YP*YP)              ! radial dist.(km)
      CLATC = dacos(YP/R)/rpd             ! geocentric colat (deg)
      ELONC = ELOND
c
c Transform field components to geocentric
c
      ALPHA= BETA-(90.d0-CLATC)*rpd       ! in radians
      CA = dcos(ALPHA)
      SA = dsin(ALPHA)
      BXC= BXD*CA - BZD*SA
      BYC= BYD
      BZC= BZD*CA + BXD*SA
      return
      end
```

```
c-----------------------------------------------------------------------
      subroutine CTODB (CLATC,ELONC, R ,BXC,BYC,BZC,
     +                  CLATD,ELOND,ALT,BXD,BYD,BZD)
c
c Geocentric to geodetic conversion for coordinates & field
c Double precision (real*8) throughout
c The root of the quartic equation is found numerically -
c EPS is computed according to the precision of the computer
c A limit, MAXIT, is placed on the number of iterations allowed
c Input (geocentric)
c    CLATC= colatitude (deg), range 0-180
c    ELONC= longitude  (degE)
c    R    = radial distance from geocentre (km)
c    BXC  = north component of field
c    BYC  = east  component of field
c    BZC  = vert  component of field
c Output (geodetic)
c    CLATD= colatitude (deg), range 0-180
c    ELOND= longitude  (degE)
c    ALT  = height above spheroid (km)
c    BXD  = north component of field
c    BYD  = east  component of field
c    BZD  = vert  component of field
c
c Colat=0 and 180 are treated as special cases
c Colat>90 is calculated from the image point in the 1st quadrant
c
      implicit real*8 (A-H,O-Z)
      rpd=datan(1.d0)/45.d0
      MAXIT=300                     ! max number of iterations allowed
      A = 6378.135d0                ! semi major axis, WGS 1972
      B = 6356.750d0                ! semi minor axis
c
c Special cases
c
      if(CLATC.lt.0.d0.or.CLATC.gt.180.d0) then       ! quit
        write(*,*)'Geocentric colat. out of range in subroutine CTODB'
        stop
      endif
      if(CLATC.eq.0.d0.or.CLATC.eq.180.d0) then       ! at poles
        CLATD=CLATC
        ELOND=ELONC
        ALT= R-B
        BXD=BXC
        BYD=BYC
        BZD=BZC
        return
      endif
c
c Evaluate EPS, the allowed error for the computed root, according to
c the precision of the computer
c
      EPS=1.d-5                     ! starting value
100   continue
      if(1.d8+EPS.gt.1.d8) then
        EPS=EPS/2.d0
        goto 100
      endif
      write(*,*)'Precision limit, EPS=',EPS
c
c Set up equations
c
       if(CLATC.le.90.d0) then             ! 1st quadrant
         IQUAD=1
         CLAT=CLATC*rpd                    ! colat in radians
       else                         ! 2nd quadrant
         IQUAD=2
         CLAT=(180.d0-CLATC)*rpd    ! image in 1st quadrant
       endif

       AA= A*A
       BB= B*B
       EE= 1.d0-(B/A)**2            ! eccentricity squared
       XP= R*sin(CLAT)              ! x coord for geodetic point
       YP= R*cos(CLAT)              ! y coord  "
       A4= EE*EE
       A3=-2.d0*XP*EE
       A2= XP*XP+(YP*YP)*BB/AA-EE*EE*AA
```

```
      A1= 2.d0*XP*AA-2.d0*XP*BB
      A0=-AA*XP*XP
c
c Find root X by Newton's method
c
      X = 0.5d0*(A+B)*sin(CLAT)   ! first approx. root
      write(*,*)
      do IT=1,MAXIT
        P1=A4*X + A3                      ! evaluate quartic
        P2=P1*X + A2
        P3=P2*X + A1
        P4=P3*X + A0
        Q1=4.d0*X + 3.d0*A3               ! evaluate derivative
        Q2=  Q1*X + 2.d0*A2
        Q3=  Q2*X +      A1
        G=-P4/Q3
        X= X+G                            ! new approximation to root
        if(mod(IT,5).eq.0) write(*,*) IT,'  Root=',X,G
        if(dabs(G).le.EPS) goto 200       ! close enough
      end do
      write(*,800) MAXIT
 800  format(' Root not found after',i4,' iterations in CTODB')
      stop
c
c Find Y
c
 200  Y = B*dsqrt(A+X)*dsqrt(A-X)/A
      if(dabs(A-X).le.10.d0*EPS) Y=0.d0
c
c Find the geodetic colatitude and longitude
c
      BETA= datan(Y/(X*(1-EE)))           ! geodetic latitude (rad)
      if(IQUAD.eq.1) then
        CLATD=90.d0-BETA/rpd              ! colat (deg) in 1st quadrant
      else
        CLATD=90.d0+BETA/rpd              ! colat (deg) in 2nd quadrant
      endif
      ELOND=ELONC
c
c Find height, positive or negative
c
      ALT = dsqrt((X-XP)**2+(Y-YP)**2)   ! km above ellipsoid
      if(YP.le.Y) ALT=-ALT         ! inside ellipsoid, negative
c
c Convert field components into geodetics
c
      ALPHA=BETA+CLAT-90.d0*rpd   ! angle between 'verticals'
      if(IQUAD.eq.1) then                 ! 1st quadrant
        BXD= BXC*dcos(ALPHA) + BZC*dsin(ALPHA)
        BZD= BZC*dcos(ALPHA) - BXC*dsin(ALPHA)
      else                                ! 2nd quadrant
        BXD= BXC*dcos(ALPHA) - BZC*dsin(ALPHA)
        BZD= BZC*dcos(ALPHA) + BXC*dsin(ALPHA)
      endif
      BYD=BYC
      return
      end

c-----------------------------------------------------------------------
      subroutine CTOMB (CLATC,ELONC,BXC,BYC,BZC,CLATP,ELONP,
     +                  CLATM,ELONM,BXM,BYM,BZM)
c
c Converts coordinates & field from geocentric to geomagnetic
c Double precision (real*8) throughout
c Radial distance from geocentre & Z-field (down) are unchanged.
c Input (geocentric)
c    CLATC= colatitude (deg), range 0-180
c    ELONC= longitude (degE), any value (+ve east)
c    BXC  = north component of field
c    BYC  = east  component of field
c    BZC  = vert  component of field
c    CLATP= pole colatitude (deg), range 0-180
c    ELONP= pole longitude (degE), any value (+ve east)
c Output (geomagnetic)
c    CLATM= colatitude (deg)
c    ELONM= longitude (degE)
c    BXM  = north component of field
c    BYM  = east  component of field
c    BZM  = vert  component of field
c
```

```
      implicit real*8 (A-H,O-Z)
      rpd=datan(1.d0)/45.d0        ! radians per degree

      if(CLATC.lt.0.d0.or.CLATC.gt.180.d0.or.
     +   CLATP.lt.0.d0.or.CLATP.gt.180.d0) then
        write(*,*)'Colatitude input out of range in subroutine CTOMB'
        stop
      endif
      CLATCR= CLATC*rpd
      CLATPR= CLATP*rpd
      ELONCR= ELONC*rpd
      ELONPR= ELONP*rpd
c
c Convert coordinates to geomagnetic (CLATM,ELONM)
c
      A= dcos(CLATPR)*dcos(CLATCR)
      B= dsin(CLATPR)*dsin(CLATCR)*dcos(ELONCR-ELONPR)
      CLATMR= dacos(A+B)
      CLATM = CLATMR/rpd                 ! magnetic colatitude
      C= dsin(CLATCR)*dsin(ELONCR-ELONPR)/dsin(CLATMR)
      ELONM= dasin(C)/rpd
      if(dcos(CLATCR).gt.dcos(CLATPR)*dcos(CLATMR)) ELONM=180.d0-ELONM
      if(ELONM.lt.0.d0) ELONM=ELONM+360.d0
c
c Convert field to geomagnetic frame
c
      SA= dsin(CLATPR)*dsin(ELONCR-ELONPR)/dsin(CLATMR)   ! sin alpha
      ALPHA=dasin(SA)/rpd                ! in deg
      if(dcos(CLATPR).lt.dcos(CLATCR)*dcos(CLATMR)) ALPHA=180.0-ALPHA
      CA  = dcos(ALPHA*rpd)              ! cos alpha
      BXM = BXC*CA - BYC*SA              ! north (mag)
      BYM = BYC*CA + BXC*SA              ! east  (mag)
      BZM = BZC
      return
      end

c-----------------------------------------------------------------
      subroutine MTOCB (CLATM,ELONM,BXM,BYM,BZM,CLATP,ELONP,
     +                  CLATC,ELONC,BXC,BYC,BZC)
c
c Converts coordinates & field from geomagnetic to geocentric
c Double precision (real*8) throughout
c Radial distance from geocentre & Z-field (down) are unchanged.
c Input (geomagnetic)
c    CLATM= colatitude (deg), range 0-180
c    ELONM= longitude (degE), any value (+ve east)
c    BXM  = north component of field
c    BYM  = east  component of field
c    BZM  = vert  component of field
c    CLATP= geomagnetic pole colatitude (deg), range 0-180
c    ELONP=     "       pole longitude (degE), any value (+ve east)
c Output (geocentric)
c    CLATC= colatitude (deg)
c    ELONC= longitude (degE)
c    BXC  = north component of field
c    BYC  = east  component of field
c    BZC  = vert  component of field
c
      implicit real*8  (A-H,O-Z)
      rpd=datan(1.d0)/45.d0              ! radians per degree

      if(CLATM.lt.0.d0.or.CLATM.gt.180.d0.or.
     +   CLATP.lt.0.d0.or.CLATP.gt.180.d0) then
         write(*,*)'Colatitude out of range in subroutine MTOCB'
         stop
      endif

      CLATMR=CLATM*rpd                   ! in radians
      CLATPR=CLATP*rpd
      ELONMR=ELONM*rpd
      ELONPR=ELONP*rpd

      A= dcos(CLATPR)*dcos(CLATMR)
      B= dsin(CLATPR)*dsin(CLATMR)*dcos(ELONMR)
      CLATCR= dacos(A-B)
      CLATC = CLATCR/rpd
      C= dsin(CLATMR)*dsin(ELONMR)/dsin(CLATCR)
      ELONC= ELONP+dasin(C)/rpd
```

```
      if(dcos(CLATMR).lt.dcos(CLATPR)*dcos(CLATCR)) then
         ELONC=180.d0+2.d0*ELONP-ELONC
      endif
      if(ELONC.lt.  0.d0) ELONC=ELONC+360.d0
      if(ELONC.gt.360.d0) ELONC=ELONC-360.d0
c
c Convert field to geocentric frame
c
      SA= dsin(CLATPR)*dsin(ELONC*rpd-ELONPR)/dsin(CLATMR)
      ALPHA=dasin(SA)/rpd                  ! in deg
      if(dcos(CLATPR).lt.dcos(CLATCR)*dcos(CLATMR)) then
        ALPHA=180.0-ALPHA                  ! obtuse case
      endif
      CA  = dcos(ALPHA*rpd)                ! cos alpha
      BXC = BXM*CA + BYM*SA                ! north (mag)
      BYC = BYM*CA - BXM*SA                ! east  (mag)
      BZC = BZM
      return
      end
c---------------------------------------------------------------------
```